Formalizing the Divergence Theorem and the Cauchy Integral Formula in Lean

Yury Kudryashov
University of Toronto at Mississauga, Canada (till July 2022)
Texas A&M University, College Station, TX, USA (since August 2022)

Abstract

I formalize a version of the divergence theorem for a function on a rectangular box that does not assume regularity of individual partial derivatives, only Fréchet differentiability of the vector field and integrability of its divergence. Then I use this theorem to prove the Cauchy-Goursat theorem (for some simple domains) and bootstrap complex analysis in the Lean mathematical library. The main tool is the GP-integral, a version of the Henstock-Kurzweil integral introduced by J. Mawhin in 1981. The divergence theorem for this integral does not require integrability of the divergence.

2012 ACM Subject Classification Security and privacy → Logic and verification; Mathematics of computing → Integral calculus

Keywords and phrases divergence theorem, Green's theorem, Gauge integral, Cauchy integral formula, Cauchy-Goursat theorem, complex analysis

Digital Object Identifier 10.4230/LIPIcs.ITP.2022.23

Supplementary Material *InteractiveResource (Documentation Website)*: `http://div-thm.urkud.name/`

Acknowledgements I want to thank Patrick Massot for bringing up an idea of formalizing a sufficiently general version of the divergence theorem, Sébastien Gouëzel for fruitful discussions and peer review of most of the code, and my wife Nataliya Goncharuk for constant support. I also want to thank Anne Baanen, Johan Commelin, Nataliya Goncharuk, and Robert Y. Lewis for valuable comments on the draft versions of this paper and I thank my son Konstantin for finding lots of typos. I am also grateful to the anonymous referees for their valuable comments.

1 Introduction

The divergence theorem says that, under certain assumptions, the integral of the divergence of a vector field over a region is equal to the flow of this vector field through the boundary of the region. For a rectangular region on the plane, this can be written as

$$\int_a^b \int_c^d \left(\frac{\partial f(x,y)}{\partial x} + \frac{\partial g(x,y)}{\partial y}\right) dydx = \int_a^b (g(x,d) - g(x,c))\, dx + \int_c^d (f(b,y) - f(a,y))\, dy, \quad (1)$$

where $f, g\colon \mathbb{R}^2 \to E$ are functions from the plane to some Banach space. This statement is also known as Green's theorem. For continuously differentiable functions f, g, the equality immediately follows from the Fundamental Theorem of Calculus and the Fubini Theorem.

If $F\colon \mathbb{C} \to E$ is a complex differentiable function, then, due to Cauchy-Riemann relations, the left-hand side of Green's theorem applied to $f(x,y) = F(x+iy)$ and $g(x,y) = iF(x+iy)$ is zero, and we get Cauchy's integral theorem (a.k.a. the Cauchy-Goursat theorem) for a rectangle.

13th International Conference on Interactive Theorem Proving (ITP 2022).
Editors: June Andronick and Leonardo de Moura; Article No. 23; pp. 23:1–23:19
Leibniz International Proceedings in Informatics
Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl Publishing, Germany

There is a gap in the proof outlined above: I explained how to prove Green's formula for a *continuously differentiable* function but the Cauchy-Goursat theorem works for any *complex differentiable* function. A common misbelief is that this makes Green's formula (and the divergence theorem) unusable for the Cauchy-Goursat theorem, and one has to prove it using an explicit infinite descent.

The goal of this project is to formulate a version of the divergence theorem that implies the Cauchy-Goursat theorem without any assumptions on the derivative of a complex differentiable function. It was originally inspired by a version of the divergence theorem by F. Acker [2] that works for a vector field with *continuous divergence*, though the actual proof I formalized is very different.

Here is the list of most important definitions and theorems I formalize in this project.

- Riemann, McShane, Henstock-Kurzweil, and GP-integrals of a function from a box in $\mathbb{R}^n$ to a real Banach space, see Sections 3 and 4;
- the divergence theorem for the GP-integral, see Theorem 2 and Section 4.9;
- the divergence theorem for the Bochner integral, see Sections 3.3 and 4.11; the Bochner integral is a generalization of the Lebesgue integral to functions that take values in Banach spaces;
- some basic theorem from complex analysis, see Sec. 5:
  - the Cauchy-Goursat theorem for rectangles, annuli, and disks;
  - the Cauchy integral formula for a disk;
  - analyticity of a complex differentiable function;
  - the Riemann removable singularity theorem;
  - maximum modulus principle;
  - Liouville's theorem;
  - Schwarz lemma.

Most of these theorems were formalized earlier in other theorem provers. However, this is the first project where the divergence theorem was formalized in this generality. In particular, the Cauchy-Goursat theorem becomes a simple corollary of the general divergence theorem. Also, the assumptions in the Cauchy-Goursat theorem are slightly weaker than in most textbooks.

The most similar other project is Abdulaziz and Paulson's formalization of Green's theorem for a large class of domains in $\mathbb{R}^2$ in Isabelle [1]. Neither Isabelle nor my version of the divergence theorem implies the other. Here are the key points where one of the formalizations is more general than the other, see also Sec. 2.

shape of the domain The Isabelle formalization works with any elementary region while my formalization only works for a rectangular box;

differentiability The Isabelle formalization only requires existence of the partial derivatives while I require Fréchet differentiability of the original function;

regularity of the derivative The Isabelle formalization requires integrability of each partial derivative while my formalization makes no regularity assumptions for the GP-integral and assumes only integrability of the divergence for the Henstock-Kurzweil and Bochner integrals; this makes it possible deduce the Cauchy-Goursat theorem from my version of the divergence theorem;

domain dimension The Isabelle formalization only deals with dimension two while I formalize the divergence theorem in any finite dimension; this generalization is needed throughout differential topology, PDE theory and mathematical physics;

codomain dimension The Isabelle formalization only works for finite dimensional codomain while my formalization works for any Banach space; this is needed, for example, in spectral theory for the resolvent.

The complex analysis part of the project does not try to compete with the state of the art complex analysis libraries in Mizar [9] and Isabelle/HOL [10], both originally written by J. Harrison, and is provided mostly as a proof of concept (and the begining of a new project).

All these theorems are already in the `master` branch of the Lean [7] mathematical library `mathlib` [11]. For presentational purposes, I changed some names (the `mathlib` naming convention sometimes leads to very long names) and notation in the code listings below.

Links to the actual formal definitions in `mathlib` are marked with the symbol ↗. The links point to a website owned by me that redirects to the code in a specific version of `mathlib`. From time to time, I will update them to point to the current `master`.

Structure of the paper

In Sec. 2 I informally discuss different ways to generalize the divergence theorem mentioned above, including comparison with the formalization in Isabelle. Then in Sec. 3 I give various definitions related to the Henstock-Kurzweil, McShane, and GP-integrals and formulate the divergence theorem for the GP-integral. In Sec. 4 I discuss some design choices I made in this project. Finally, in Sec. 5 I explain how to apply this theorem to prove the Cauchy-Goursat theorem and some other basic theorems from complex analysis. Sec. 6 is devoted to my future plans.

2 Generalizations of the divergence theorem

The divergence theorem for continuously differentiable vector fields on rectangular boxes can be generalized in a few different directions, leading to several theorems, none of which implies the others.

2.1 Shape of the domain

One natural direction of generalization is to deal with non-rectangular domains. This generalization is done by Abdulaziz and Paulson in Isabelle [1] for regions that can be divided both into type I regions by vertical lines only and into type II regions by horizontal lines only. I only deal with rectangular boxes, so I am not going to discuss this in any more details.

2.2 More general codomain

Another possible direction of generalization is to deal with functions $f\colon \mathbb{R}^n \to E^n$, where E is a real Banach space, instead of vector fields $v\colon \mathbb{R}^n \to \mathbb{R}^n$ and E^n is the direct sum of n copies of E. Most proofs that work for a vector field can be generalized to this case with no or little modifications. I deal with functions $\mathbb{R}^n \to E^n$ right away.

2.3 Integrable partial derivatives

It is easy to see that the standard proof based on Fubini's theorem and FTC works for a function $f\colon \mathbb{R}^n \to E^n$ such that the partial derivatives $\frac{\partial f^i(x)}{\partial x^i}$ exist at all points of the box and are integrable on the whole box.

▶ Remark 1. Here and below I use upper indices for coordinates of vectors to disambiguate i-th coordinate of a vector from i-th element in a sequence of vectors. I do not assume implicit summation over indices that appear both as an upper and as a lower index.

Neither this generalization nor the next one imply each other. I formalized the other one because it implies the Cauchy–Goursat theorem.

2.4 Fréchet differentiability

A function $f\colon E_1 \to E_2$ between normed vector spaces is called *Fréchet differentiable* at a point x with the derivative given by a continuous linear map $f'\colon E_1 \to E_2$ if $f(x+y) = f(x) + f'(y) + o(y)$ as $y \to 0$. A Fréchet differentiable function f has all partial derivatives and they are equal to the values of f' on the basis vectors.

In 1981, Mawhin [12] introduced a generalization of the Henstock-Kurzweil integral he called the *GP-integral* (probably from *generalized Perron*). This integral allows us to prove the following theorem.

▶ **Theorem 2** (see [12]). *Let E be a real Banach space. Let $f\colon \mathbb{R}^n \to E$ be a function that is Fréchet differentiable at all points of a closed box $\overline{I} = [a, b]$.*

Then for each $i = 1, \dots, n$, the partial derivative $\frac{\partial f}{\partial x^i}$ is GP-integrable on I and its integral is equal to the difference of the integrals of f over the faces $x^i = b^i$ and $x^i = a^i$ of I.

In particular, for a function $f\colon \mathbb{R}^n \to E^n$, the divergence

$$\operatorname{div} f = \sum_{i=1}^{n} \frac{\partial f^i}{\partial x^i} \tag{2}$$

is GP-integrable on I and its integral is equal to the sum of integrals of f over the faces of I with appropriate signs.

Compared to the previous generalization, this one requires differentiability in a stronger sense (Fréchet differentiability instead of existence of the partial derivatives) but it requires less regularity from the derivatives (no requirements for the GP-integral and integrability of the divergence instead of integrability of partial derivatives for most other integrals).

2.5 Weaker assumptions on a subset of the box

One can push arguments in the proofs discussed above to weaken the regularity assumptions on a "small" subset of the box.

I prove that Theorem 2 works even if on some countable subset of the box the function is continuous, not differentiable. It is possible to push these arguments even further (especially if we use another generalization of the Henstock-Kurzweil integral) but I did not formalize these theorems (yet).

3 Henstock-Kurzweil integral: informal description

While my main goal was to prove the divergence theorem for the Bochner integral, I first proved the divergence theorem for the GP-integral, see below. This integral is one of possible generalizations of the Henstock-Kurzweil integral to higher dimension. In this section, I state the relevant definitions and theorems.

3.1 Henstock-Kurzweil integral in dimension one

I start with the definition of the one-dimensional Henstock-Kurzweil integral.

▶ **Definition 3.** *We say that a tagged partition* $a = x_0 < x_1 < \cdots < x_k = b$ *with tags* $\xi_i \in [x_i, x_{i+1}]$, $0 \leqslant i < k$, *is* subordinate *to a function* $\delta\colon [a,b] \to \mathbb{R}$ *if* $[x_i, x_{i+1}] \subset [\xi_i - \delta(\xi_i), \xi_i + \delta(\xi_i)]$ *for all* $0 \leqslant i < k$.

We say that $f\colon \mathbb{R} \to E$ *has the Henstock-Kurzweil integral* c *over an interval* $[a,b]$, $\int_a^b f(x)\,dx = c$, *if for any* $\varepsilon > 0$ *there exists an everywhere positive function* $\delta\colon [a,b] \to \mathbb{R}$ *such that for any tagged partition* $(\{x_i\}, \{\xi_i\})$ *subordinate to* δ, *the Riemann sum* $\sum_{i=0}^{k-1} f(\xi_i)(x_{i+1} - x_i)$ *is* ε*-close to* c.

Slight modifications of Definition 3 give the integrals of Riemann (we require δ to be a constant) and McShane (we no longer require that tags belong to their intervals but still require $\xi_i \in [a,b]$).

An equivalent (but more complicated) definition was given by A. Denjoy to integrate $\frac{1}{x}\sin\left(\frac{1}{x^3}\right)$ over $[-1,1]$. This function is instegrable in the sense of the Henstock-Kurzweil integral but it is not Lebesgue integrable.

A well-known property of the Henstock-Kurzweil integral is the following form of the Fundamental Theorem of Calculus.

▶ **Theorem 4.** *Let* E *be a real Banach space. Let* $f\colon \mathbb{R} \to E$ *be a function differentiable on an interval* $[a,b]$. *Then its derivative is Henstock-Kurzweil integrable on* $[a,b]$ *and the integral is equal to* $f(b) - f(a)$.

The one-dimensional Henstock-Kurzweil integral is not a part of `mathlib` and I decided not to formalize it. When we need it, it will be easy to define it as a thin wrapper on top of the Henstock-Kurzweil integral over a box in $\mathbb{R}^1$ `= fin 1 → ℝ`.

3.2 Henstock-Kurzweil integral in higher dimension

We will need a few definitions to generalize the Henstock-Kurzweil integral to functions $f\colon \mathbb{R}^n \to E$.

▶ **Definition 5.** *For* $a, b \in \mathbb{R}^n$, *the* open box (a,b) *is the product of open intervals* (a^i, b^i), *that is* $(a,b) = \{x \in \mathbb{R}^n \mid \forall i, x^i \in (a^i, b^i)\}$. *The* open-closed box $(a,b]$ *and the* closed box $[a,b]$ *are defined in a similar way.*

The distortion *(or* irregularity*)* $\sigma(I)$ *of an open-closed box* $I = (a,b]$ *is the maximum of the ratios* $\frac{b^i - a^i}{b^j - a^j}$ *over all* i, j.

A partition *of an open-closed box* $I = (a,b]$ *is a finite collection of pairwise disjoint boxes* $J_k = (a_k, b_k]$ *that cover* I.

A tagged partition *of an open-closed box* $I = (a,b]$ *is a partition* $\{J_k\}$ *together with a collection of points called* tags $\xi_k \in [a,b]$, *one per each box of the partition.*

A Henstock tagged partition *is a tagged partition such that each tag belongs to the corresponding closed box:* $\xi_k \in [c_k, d_k]$, *where* $J_k = (c_k, d_k]$.

A tagged partition $\{(J_k, \xi_k)\}$ *of a box* I *is* subordinate *to a* gauge function $\delta\colon \mathbb{R}^n \to \mathbb{R}_{>0}$ *if each box* J_k *is included by the closed box* $B_{\delta(\xi_k)}(\xi_k) = [\xi_k - \delta(\xi_k), \xi_k + \delta(\xi_k)]$.

Now we are ready to define the Henstock-Kurzweil integral of a function $f\colon \mathbb{R}^n \to E$ over a box I.

▶ **Definition 6.** *Let E be a real Banach space. We say that a function $f\colon \mathbb{R}^n \to E$ is* Henstock-Kurzweil integrable *on I with integral c if for any $\varepsilon > 0$ there exists a gauge function $\delta\colon \mathbb{R}^n \to \mathbb{R}_{>0}$ such that for any Henstock tagged partition $\{ (J_k, \xi_k) \}$ of I that is subordinate to δ, the Riemann sum of f over this partition is ε-close to c.*

Similarly to the one-dimensional case, small modifications of Definition 6 provide definitions of Riemann and McShane integrability, see the paragraph after Definition 3.

Divergences of some vector fields are not integrable in the sense of this definition (and any other generalization of the Henstock-Kurzweil integral that satisfies Fubini's theorem, as shown by W. Pfeffer [13]), so we need another generalization to prove the divergence theorem.

There are several generalizations of the Henstock-Kurzweil integral such that the divergence of any Fréchet differentiable function is integrable, see Bongiorno's survey [6]. I use the one suggested by Mawhin [12].

▶ **Definition 7.** *The* distortion *(or* irregularity*) $\Sigma(\pi)$ of a partition $\pi = \{ J_k \}$ is the maximum of the distortions $\sigma(J_k)$ of the boxes J_k.*

A function $f\colon \mathbb{R}^n \to E$ is GP-integrable *on a box I with integral c if for any $\varepsilon > 0$ and a real number d there exists a gauge function $\delta\colon \mathbb{R}^n \to \mathbb{R}_{>0}$ such that for any Henstock tagged partition $\{ (J_k, \xi_k) \}$ of I that is subordinate to δ and has a distortion less than d, the Riemann sum of f over this partition is ε-close to c.*

I formalized this definition (together with a few other definitions of "box" integrals) and a proof of Theorem 2.

In dimension two, this theorem implies (1) for any pair of functions $f, g\colon \mathbb{R}^2 \to E^2$ Fréchet differentiable on a closed rectangle.

3.3 Application to the Bochner integral

The mathlib project uses the Bochner integral as its main integral [14]. This integral is a generalization of the Lebesgue integral. It was introduced by Bochner in [4], and it is also formalized in Coq [5] and Isabelle [3].

A Bochner integrable function on a box in $\mathbb{R}^n$ is McShane integrable (hence Henstock-Kurzweil and GP-integrable), thus Theorem 2 holds for the Bochner integral if we assume that the divergence is Bochner integrable on the box. Taking the limit over an exhaustion of the box by smaller boxes, one can generalize the theorem to functions that are differentiable in the interior of a box and continuous on the whole box. Here is the precise statement of the divergence theorem for Bochner integral.

▶ **Theorem 8.** *Let E be a real Banach space. Let $I = (a, b] \subset \mathbb{R}^n$ be an open-closed box. Let $s \subset \mathbb{R}^n$ be a countable set. Let $f\colon \mathbb{R}^n \to E^n$ be a function that is continuous on $\overline{I}$ and is Fréchet differentiable at all points of $\mathring{I} \setminus s$. Assume that the divergence* $\operatorname{div} f$ *defined by* (2) *is Bochner integrable on I. Then its integral is equal to the sum of integrals of f over the faces of $\overline{I}$ taken with appropriate signs (plus for the integral over a front face $x^i = b^i$ and minus for the integral over a back face $x^i = a^i$).*

4 Divergence theorem: design choices and implementation details

The code below uses some notation that is specific either to `mathlib` or this project.

`Icc a b`	the closed interval $[a, b]$;
`Ioo a b`	the open interval (a, b);
`Ioc a b`	the open-closed interval $(a, b]$;
`ICC a b`	the unordered closed interval $[\min(a, b), \max(a, b)]$;
`IOO a b`	the unordered open interval $(\min(a, b), \max(a, b))$;
$\mathbb{R}^n$	the vector space $\mathbb{R}^n$, implemented as `fin n → ℝ`, where `fin n` is the canonical type with n elements;
`E`^{n}	the direct sum of n copies of a vector space E, implemented as `fin n → E`; if $E = \mathbb{R}$, then this notation agrees with the previous one;
`ℝ>0, ℝ≥0`	the types of positive and nonnegative real numbers, respectively;
`e i`	i-th basis vector in $\mathbb{R}^n$;
i	the imaginary unit;
`s ×ℂ t`	the product of sets on the real and imaginary axes in $\mathbb{C}$.

4.1 Boxes

I use open-closed boxes in $\mathbb{R}^n$ as elements of partition because, this way, the boxes are disjoint as sets and cover the whole ambient box, so one does not have to deal with interiors or closures to define a partition.

```
structure box (n : ℕ) : Type :=
(lower upper : ℝⁿ)
(lower_lt_upper : ∀ i, lower i < upper i)
```

Each box can be interpreted as a set in $\mathbb{R}^n$.

```
instance : has_mem (ℝⁿ) (box n) :=
⟨λ x I, ∀ i, x i ∈ Ioc (I.lower i) (I.upper i)⟩

instance : has_coe_t (box n) (set} ℝⁿ) := ⟨λ I, {x | x ∈ I}⟩
```

The order on the boxes is the inclusion order on the corresponding sets.

I chose to explicitly deny empty boxes because this way the `lower` and `upper` vertices are uniquely defined by the set of points that belong to the box. The empty box is represented as the bottom element `⊥ : with_bot (box n)`, where `with_bot` α is the type `{⊥} ∪` α, implemented as a type synonym for `option` α with custom order.

From the order theory point of view, the type `with_bot (box n)` is a lattice, where the meet of two boxes is their intersection and the join of two boxes is the minimal box that includes both of them.

For an open-closed box `I : box n`, `I.Ioo` and `I.Icc` are the corresponding open and closed boxes, respectively.

Given a box `I : box (n + 1)` and an index `i : fin (n + 1)`, I define the i-th *face* of I to be the box in $\mathbb{R}^n$ with lower and upper vertices given by `I.lower ∘ i.succ_above` and `I.upper ∘ i.succ_above`, where `i.succ_above : fin n → fin (n + 1)` is the unique monotone embedding leaving a "hole" at i: it sends $j < i$ to j and $j \geqslant i$ to $j + 1$. I also define two embeddings `I.front_face i, I.back_face i :` $\mathbb{R}^n \to \mathbb{R}^{n+1}$ that insert `I.upper i` and `I.lower i`, respectively, as the i-th coordinate. These embeddings map `I.face i` to the faces `x i = I.upper i` and `x i = I.lower i` of I.

4.2 Partitions

A *partition* of a box I is a finite collection of pairwise disjoint boxes that cover I. Sometimes, it is useful to deal with a collection of pairwise disjoint boxes that cover only a part of I, so I define a *prepartition* of a box and a predicate `is_partition` saying that a prepartition is actually a partition.

```
structure prepartition (I : box n) : Type :=
(boxes : finset (box n))
(le_of_mem' : ∀ J ∈ boxes, J ≤ I)
(pairwise_disjoint : pairwise boxes
  (disjoint on (coe : box n → set ℝⁿ)))

def is_partition (π : prepartition I) : Prop :=
∀ x ∈ I, ∃ J ∈ π, x ∈ J
```

I do not use two separate structures `prepartition` and `partition` because with my approach I can define a function of prepartitions (e.g., the Riemann integral sum), then prove statements about its limits along various filters; some of these filters require a prepartition to be a partition, some do not. Also, I can define operations on prepartitions, then freely mix partitions and prepartitions in the arguments instead of adding an explicit `to_prepartition` or an implicit coercion here and there. On the other hand, I have to use `(π : prepartion I) (hπ : is_partition π)` instead of `(π : partition I)` whenever I want to argue about a partition, so I am not completely sure that I made the right choice.

I establish basic API about (pre)partitions, most notably the following predicates, relations, and operations.

- $\pi_1 \leqslant \pi_2$: we say that one prepartition is less than or equal to another if each box of the former is included in some box of the latter; with this order, we get a bounded meet-semilattice structure on prepartitions of a box;
- `Union`: the union of all boxes of a prepartition, as a set in $\mathbb{R}^n$; for a partition, this union is equal to the original box;
- `bUnion`: given a prepartition π of I and a function π' that sends each box in $\mathbb{R}^n$ to a prepartition of that box, returns the prepartition of I formed by the boxes of $\pi' J$, $J \in \pi$;
- `split_center`: the partition of a box into 2^n subboxes by the coordinate hyperplanes passing through the center of the box;
- `split`: the partition of a box into two subboxes by a coordinate hyperplane (or the trivial one-box partition if the hyperplane does not meet the box);
- `split_many`: the partition of a box into subboxes by a finite set of coordinate hyperplanes;
- `compl`: an unspecified prepartition such that `π.compl.Union = I \ π.Union`; I prove that it exists, then use the axiom of choice to get a witness.

4.3 Tagged partitions

Recall that a *tagged (pre)partition* is a (pre)partition π with a point ("tag") chosen in each box of π. In the formal definition I require that the `tag` function is defined on all boxes. This way I can write `π.tag J` without proving that J is one of the boxes of π.

```
structure tagged_prepartition (I : box n) extends prepartition I :=
(tag : box n → ℝⁿ)
(tag_mem_Icc : ∀ J, tag J ∈ I.Icc)
```

Here the extends keyword implicitly adds a field `to_prepartition` to the structure and introduces some syntatic sugar for composed projections (e.g., given π of type `tagged_prepartition I`, Lean unfolds `π.boxes` to `π.to_prepartition.boxes`) and constructors.

Unfortunately, similar syntax does not work for other definitions and lemmas in the same namespace, so I have to repeat some definitions and lemmas about prepartition once more (with one line proofs that reference the corresponding lemma about prepartitions). I hope that this tedious work will be automized in a future version of Lean or mathlib.

There are a few new definitions about tagged prepartitions that essentially use the tags, see Definition 5.

```
def is_Henstock (π : tagged_prepartition I) : Prop :=
∀ J ∈ π, π.tag J ∈ J.Icc

def is_subordinate (π : tagged_prepartition I) (r : ℝⁿ → ℝ>0) : Prop :=
∀ J ∈ π, J.Icc ⊆ closed_ball (π.tag J) (r (π.tag J))
```

4.4 Cousin's lemma

Cousin's lemma says that for any gauge function $\delta\colon \mathbb{R}^n \to \mathbb{R}_{>0}$ and a box, there exists a tagged partition of this box that is subordinate to δ. This lemma is needed to show that the Henstock-Kurzweil integral is well defined: if it was false, any number would be the Henstock-Kurzweil integral of any function.

I prove two versions of this lemma. First I prove the lemma as stated in the previous paragraph with an additional assertion that all boxes of the partition are homothetic to the original box.

```
lemma exists_subordinate_Henstock (I : box n) (r : ℝⁿ → ℝ>0) :
  ∃ π : tagged_prepartition I, π.is_partition ∧ π.is_Henstock ∧
    π.is_subordinate r ∧
    (∀ J ∈ π, ∃ m : ℕ, ∀ i, J.upper i - J.lower i =
      (I.upper i - I.lower i) / 2 ^ m) ∧
    π.distortion = I.distortion
```

Here is the sketch of the proof. If a box I does not admit a tagged partition with these properties, then the same is true for one of the 2^n boxes of the partition `I.split_center`. Thus we obtain an infinite sequence of boxes J_k, each one is twice smaller than the previous one in each direction, such that none of these boxes admit a partition with these properties. Let a be the unique common point of these boxes. For sufficiently large k, J_k is included in the closed ball with center a and radius $r(a)$, hence the one-box partition of J_k with tag at a satisfies all the required properties. This contradiction proves the lemma.

Then I use it to prove that for any prepartition π there exists a refinement of this prepartition with the same distortion, see Definition 7, and a choice of tags such that the resulting tagged prepartition is Henstock and is subordinate to a given gauge function. To prove this, I apply the previous lemma to each box of π, then merge these partitions using `prepartition.bUnion`. I use this version of Cousin's lemma to prove that the GP-integral is well defined.

```
lemma exists_le_Henstock_Union_eq (r : ℝⁿ → ℝ>0) (π : prepartition I) :
  ∃ π' : tagged_prepartition I, π'.to_prepartition ≤ π ∧
    π'.is_Henstock ∧ π'.is_subordinate r ∧ π'.distortion = π.distortion ∧
    π'.Union = π.Union :=
```

4.5 The filters

Different "box" integrals (Riemann, McShane, Henstock-Kurzweil, GP) are defined as the limits of the Riemann sums along some filters on the space of partitions of a box.

I define the structure `integration_params` that holds data needed to define either of these four integrals and a few more.

```
structure integration_params : Type :=
(bRiemann bHenstock bDistortion : bool)
```

The parameters have the following meaning:

- `bRiemann` this is a Riemann integral, the gauge function must be a constant;
- `bHenstock` tags must belong to the closure of their boxes;
- `bDistortion` the gauge function may depend on the distortion of a partition.

The integration parameters used to define the Riemann, Henstock-Kurzweil, McShane, and GP-integrals are listed below.

```
def Riemann : integration_params :=
{ bRiemann := tt,
  bHenstock := tt,
  bDistortion := ff }

def Henstock : integration_params :=
{ bRiemann := ff,
  bHenstock := tt,
  bDistortion := ff }

def McShane : integration_params :=
{ bRiemann := ff,
  bHenstock := ff,
  bDistortion := ff }

def GP : integration_params :=
{ bRiemann := ff,
  bHenstock := tt,
  bDistortion := tt }
```

On one hand, this design choice allows me to prove some lemmas (e.g., Henstock-Sacks inequality, see Sec. 4.8) uniformly for all these integrals. On the other hand, it is hard to add more integrals to the collection, see Bongiorno's survey [6] for other reasonable generalizations of the Henstock-Kurzweil integral to higher dimension.

I require that the gauge function satisfies the following condition.

```
def r_cond {n : ℕ} (l : integration_params) (r : ℝⁿ → ℝ>0) : Prop :=
l.bRiemann → ∀ x, r x = r 0
```

If `l.bRiemann` is false, then this condition is trivial, otherwise it says that the gauge function is actually a constant function.

For each set of parameters `l : integration_params` and a box I, we define several filters on the space of prepartitions of I. All these filters have basis sets of the same type.

```
structure mem_base_set (l : integration_params) (I : box n) (c : ℝ≥0)
  (r : ℝⁿ → ℝ>0) (π : tagged_prepartition I) : Prop :=
(is_subordinate : π.is_subordinate r)
(is_Henstock : l.bHenstock → π.is_Henstock)
```

```
(distortion_le : l.bDistortion → π.distortion ≤ c)
(exists_compl : l.bDistortion → ∃ π' : prepartition I,
  π'.Union = I \ π.Union ∧ π'.distortion ≤ c)
```

We already saw the first three assumptions in the informal definition of the GP-integral. The last assumption is trivial if π is a partition (then we can choose π' to be the empty prepartition). One can show that it is also trivial whenever $c > 1$ but I decided to cut a corner here and introduce this assumption instead of adding an explicit assumption $c > 1$ here and there and proving one more lemma.

The most important filter related to a set of integration parameters is the filter `to_filter_Union` below. Cousin's lemma from Sec. 4.4 implies that this filter is non-trivial. The actual definition uses more intermediate filters but the result is equal (though not definitionally equal) to the code below.

```
def to_filter (l : integration_params) (I : box n) :
  filter (tagged_prepartition I) :=
⨆ c : ℝ≥0, ⨅ (r : ℝⁿ → ℝ>0) (hr : l.r_cond r),
  𝓟 {π | l.mem_base_set I c r π}

def to_filter_Union (l : integration_params) (I : box n)
  (π₀ : prepartition I) :=
l.to_filter I ⊓ 𝓟 {π | π.Union = π₀.Union}
```

The filter `to_filter` is useful to prove facts about integrals on subboxes.

4.6 Box-additive function

I introduce the following definition.

▶ **Definition 9.** *A function on boxes in* $\mathbb{R}^n$ *taking values in an additive group is said to be* box-additive *if for any box* I *and its partition* π*, the sum of its values on the boxes of* π *is equal to its value on* I*.*

A function is said to be box-additive on subboxes of I_0 *if the same property holds true whenever* $I \leqslant I_0$*.*

In order to deal with these two notions simultaneously, the actual definition takes an argument of the type `with_top (box n)`. Similarly to the type `with_bot (box n)`, this type is the disjoint union of `box n` and the top element $\top$.

```
structure box_additive_map (n : ℕ) (G : Type*) [add_comm_group G]
  (I : with_top (box n)) : Type* :=
(to_fun : box n → G)
(sum_partition_boxes' : ∀ J : box n, (J : with_top (box n)) ≤ I →
  ∀ π : prepartition J, π.is_partition →
    ∑ Ji in π.boxes, to_fun Ji = to_fun J)
```

I use notation `n` $\to^{ba}$ `G` for functions that are box-additive on the whole space and `n` $\to^{ba}$`[I] G` for functions that are box-additive on subboxes of I.

It suffices to verify additivity only on the two-box partitions `split` introduced above. Indeed, let t be the set of all hyperplanes that contain faces of a partition π of I. Then the partition `split_many I t` can be obtained by a series of splits along a single hyperplane both from the trivial one-box partition and from π. Therefore, if f is additive on the two-box partitions, then the sum of values of f over all boxes of `split_many I t` is equal both to $f(I)$ and to the sum of its values over all boxes of π.

```
def of_map_split_add (f : box n → G) (I₀ : with_top (box n))
  (hf : ∀ I : box n, (I : with_top (box n)) ≤ I₀ →
    ∀ {i x}, x ∈ Ioo (I.lower i) (I.upper i) →
    f (I.split_lower i x) + f (I.split_upper i x) = f I) :
  n →ᵇᵃ[I₀] G
```

Here `I.split_lower i x` and `I.split_upper i x` are the boxes of the partition `split I`. Since one of them can be empty, they have type `with_bot (box n)`, so the actual code looks like `option.elim (I.split_lower i x) 0 f` instead of `f (I.split_lower i x)`.

Each locally finite measure μ defines a box-additive function. Next, if $f : \mathbb{R}^n \to E$ is integrable (in any reasonable sense) on a box I, then its integral over a box is a box-additive function on subboxes of I, see Sec. 4.8.

One more construction of box-additive functions appears in the proof of the divergence theorem. Given a box-additive function f_y on each cross-section $x^i = y$, $y \in [a^i, b^i]$ of a closed box `I₀.Icc = [a, b]`, the function given by `g J = f (J.upper i) (J.face i) - f (J.lower i) (J.face i)` is box additive on subboxes of I_0.

To ensure nice definitional equality properties of the result, the actual definition involves two functions and a proof that they are equal on all the boxes relevant to the definition.

```
def upper_sub_lower {G : Type u} [add_comm_group G]
  (I₀ : box (n + 1)) (i : fin (n + 1)) (f : ℝ → box n → G)
  (fb : Icc (I₀.lower i) (I₀.upper i) → n →ᵇᵃ[I₀.face i] G)
  (hf : ∀ x (hx : x ∈ Icc (I₀.lower i) (I₀.upper i)) J,
    f x J = fb ⟨x, hx⟩ J) :
  (n + 1) →ᵇᵃ[I₀] G :=
```

4.7 Box integral

The box integral of a function $f : \mathbb{R}^n \to E$ on an open-closed box I in the sense of integration parameters l is defined as the limit of the Riemann sum over a partition π of I along the filter `l.to_filter_Union ⊤`, where `⊤` is the one-box partition of I. If the limit does not exist, then the integral is defined to be zero.

I define the integral of a function $f : \mathbb{R}^n \to E$ with respect to a box-additive volume taking values in the space of continuous linear functions `E →L[ℝ] F`. This way one can use the same definition, e.g., for Riemann-Stieltjes integrals. However, I only used this definition in the case $E = F$ and `vol J x = (μ J).to_real · x` for some measure μ. So, this generalization might be a case of overengineering.

```
def integral_sum (f : ℝⁿ → E) (vol : n →ᵇᵃ (E →L[ℝ] F))
  (π : tagged_prepartition I) : F :=
∑ J in π.boxes, vol J (f (π.tag J))

def has_integral (I : box n) (l : integration_params) (f : ℝⁿ → E)
  (vol : n →ᵇᵃ (E →L[ℝ] F)) (y : F) : Prop :=
tendsto (integral_sum f vol) (l.to_filter_Union I ⊤) (𝓝 y)

def integrable (I : box n) (l : integration_params) (f : ℝⁿ → E)
  (vol : n →ᵇᵃ (E →L[ℝ] F)) : Prop :=
∃ y, has_integral I l f vol y
```

```
def integral (I : box n) (l : integration_params) (f : ℝⁿ → E)
  (vol : n →ᵇᵃ (E →L[ℝ] F)) : F :=
if h : integrable I l f vol then h.some else 0
```

Usual theorems (uniqueness of the integral, Cauchy convergence test, additivity) immediately follow from the definition and properties of the limit.

4.8 The Henstock-Sacks inequality

The Henstock-Sacks inequality for the Henstock-Kurzweil integral says the following. Let f be a function integrable on a box I; let $\delta\colon \mathbb{R}^n \to \mathbb{R}_{>0}$ be a gauge function such that for any tagged partition of I subordinate to δ, the integral sum over this partition is ε-close to the integral. Then for any tagged *prepartition* π, the integral sum over π differs from the integral of f over the part of I covered by π by at most ε.

This inequality is used, e.g., to prove that a function that is Henstock-Kurzweil integrable on a box I, is Henstock-Kurzweil integrable on any subbox of I and defines a box-additive function on subboxes of I. I prove several versions of this inequality for any of the "box" integrals.

Instead of using predicate assumptions on δ, I define `convergence_r` to be a function $\delta\colon \mathbb{R}^n \to \mathbb{R}_{>0}$ such that

- if `l.bRiemann`, then δ is a constant;
- if $\varepsilon > 0$, then for any tagged partition π of I satisfying the predicate `l.mem_base_set I c δ`, the integral sum of f over π differs from the integral of f over I by at most ε.

```
def convergence_r (h : integrable I l f vol) (ε : ℝ) (c : ℝ≥0) :
  ℝⁿ → ℝ>0
```

Let me quote two versions of the Henstock-Sacks inequality here. One version compares the Riemann sums of a function over two prepartitions that cover the same part of the box.

```
lemma dist_integral_sum_le_of_mem_base_set (h : integrable I l f vol)
  (hpos₁ : 0 < ε₁) (hpos₂ : 0 < ε₂)
  (hπ₁ : l.mem_base_set I c₁ (h.convergence_r ε₁ c₁) π₁)
  (hπ₂ : l.mem_base_set I c₂ (h.convergence_r ε₂ c₂) π₂)
  (hU : π₁.Union = π₂.Union) :
  dist (integral_sum f vol π₁) (integral_sum f vol π₂) ≤ ε₁ + ε₂
```

The other version replaces one of these Riemann sums with the sum of integrals of the function over the boxes of the partition.

```
lemma dist_integral_sum_sum_integral_le_of_mem_base_set_of_Union_eq
  (h : integrable I l f vol) (hpos : 0 < ε)
  (hπ : l.mem_base_set I c (h.convergence_r ε c) π)
  (hU : π.Union = π₀.Union) :
  dist (integral_sum f vol π) (∑ J in π₀.boxes, integral J l f vol) ≤ ε
```

4.9 Divergence theorem for the GP-integral

To prove the divergence theorem for the GP-integral Theorem 2, we prove that each partial derivative is GP-integrable with the integral equal to the difference of the integral of the original function over the front and back faces of the box.

```
lemma has_integral_GP_pderiv (f : ℝⁿ⁺¹ → E) (f' : ℝⁿ⁺¹ → ℝⁿ⁺¹ →L[ℝ] E)
  (s : set (ℝⁿ⁺¹)) (hs : countable s)
  (Hs : ∀ x ∈ s, continuous_within_at f I.Icc x)
  (Hd : ∀ x ∈ I.Icc \ s, has_fderiv_within_at f (f' x) I.Icc x)
  (i : fin (n + 1)) :
  has_integral I GP (λ x, f' x (e i)) volume
    (integral (I.face i) GP (f ∘ I.front_face i) volume -
      integral (I.face i) GP (f ∘ I.back_face i) volume)
```

Here and in the next listing, *volume* is a notation for the box-additive volume `volume : (n + 1) →ᵇᵃ (E →L[ℝ] E)` defined by `dV J x = (∏ j, J.upper j - J.lower j) · x`.

The sum of these statements for all terms of the divergence gives us the divergence theorem.

```
lemma has_integral_GP_divergence (f : ℝⁿ⁺¹ → Eⁿ⁺¹)
  (f' : ℝⁿ⁺¹ → ℝⁿ⁺¹ →L[ℝ] Eⁿ⁺¹) (s : set ℝⁿ⁺¹) (hs : countable s)
  (Hs : ∀ x ∈ s, continuous_within_at f I.Icc x)
  (Hd : ∀ x ∈ I.Icc \ s, has_fderiv_within_at f (f' x) I.Icc x) :
  has_integral I GP (λ x, ∑ i, f' x (e i) i) volume
    (∑ i, (integral (I.face i) GP (f ∘ I.front_face i) volume -
      integral (I.face i) GP (f ∘ I.back_face i) volume))
```

The proof of the main lemma follows the same scheme as the standard proof of the Fundamental Theorem of Calculus for the Henstock-Kurzweil integral: given a positive number ε and an upper estimate c on the distortion of the partition, for each point x of differentiability, one can choose $\delta(x) > 0$ such that the estimate $f(y) = f(x) + f'(x)(y - x) + o(y - x)$ and $\Sigma(\pi) \leqslant c$ imply that the difference between the integrals of f over the i-th front and back faces of a $\delta(x)$-small box $J \ni x$ is $\varepsilon V(J)$-close to the term $\frac{\partial f}{\partial x^i} V(J)$ of the Riemann sum for the integral of the partial derivative.

I do one modification to this argument that allows me to use weaker assumptions on a countable set of points. Namely, for $x \in s$ I choose $\delta(x)$ so that for a $\delta(x)$-small box $J \ni x$, the term corresponding to this box has norm less than $\kappa(x) > 0$, where $\sum_{x \in s} \kappa(x) < \varepsilon/2$. This is possible due to the continuity of f at x.

4.10 McShane and Bochner integrability

In order to transfer the result from the GP-integral to the Bochner integral, I prove that any Bochner integrable function is integrable in the sense of any box integral with `bRiemann = ff`. In other words, a Bochner integrable function is integrable in the sense of the McShane, Henstock-Kurzweil, and GP-integrals.

```
lemma integrable_on.has_box_integral {f : ℝⁿ → E} {μ : measure ℝⁿ}
  [is_locally_finite_measure μ] {I : box n} (hf : integrable_on f I μ)
  (l : integration_params) (hl : l.bRiemann = ff) :
  has_integral I l f μ.to_box_additive.to_smul (∫ x in I, f x ∂μ)
```

The proof follows R. Gordon's book [8], with some modifications required to generalize it from a function $f : \mathbb{R} \to \mathbb{R}$ to a function $f : \mathbb{R}^n \to E$.

First, I prove that the indicator function of a measurable set s is McShane integrable with respect to the volume defined by a locally finite measure μ. This immediately implies integrability of simple functions. Then I show that changing the values of a function on a set of measure zero does not affect its integral. Finally, I use approximations of integrable functions by a sequence of simple functions to prove McShane integrability of a Bochner integrable function.

4.11 Divergence theorem for the Bochner integral

Since any Bochner integrable function is GP-integrable, we immediately obtain the divergence theorem for the Bochner integral under assumptions that the function is continuous on a closed box, is differentiable at all but countably many points of this box, and has integrable divergence.

```
lemma integral_divergence_aux (I : box (n + 1)) (f : ℝⁿ⁺¹ → Eⁿ⁺¹)
  (f' : ℝⁿ⁺¹ → ℝⁿ⁺¹ →L[ℝ] Eⁿ⁺¹) (s : set ℝⁿ⁺¹) (hs : countable s)
  (Hc : continuous_on f I.Icc)
  (Hd : ∀ x ∈ I.Icc \ s, has_fderiv_within_at f (f' x) I.Icc x)
  (Hi : integrable_on (λ x, ∑ i, f' x (e i) i) I.Icc) :
  ∫ x in I.Icc, ∑ i, f' x (e i) i =
    ∑ i : fin (n + 1),
      ((∫ x in (I.face i).Icc, f (I.front_face i x) i) -
        ∫ x in (I.face i).Icc, f (I.back_face i x) i)
```

I slightly generalize this result. First, I drop the differentiability assumption on the boundary of I. To do this, I apply the auxiliary result to an increasing sequence of subboxes that cover the interior of I. Second, I allow $a^i = b^i$; in this case both sides are equal to zero.

```
lemma integral_divergence (a b : ℝⁿ⁺¹) (hle : a ≤ b) (f : ℝⁿ⁺¹ → Eⁿ⁺¹)
  (f' : ℝⁿ⁺¹ → ℝⁿ⁺¹ →L[ℝ] Eⁿ⁺¹) (s : set ℝⁿ⁺¹) (hs : countable s)
  (Hc : continuous_on f (Icc a b))
  (Hd : ∀ x : ℝⁿ⁺¹, (∀ i, x i ∈ Ioo (a i) (b i)) → x ∉ s,
    has_fderiv_at f (f' x) x)
  (Hi : integrable_on (λ x, ∑ i, f' x (e i) i) (Icc a b)) :
  ∫ x in Icc a b, ∑ i, f' x (e i) i =
    ∑ i : fin (n + 1), ((∫ x in face i, f (front_face i x) i) -
      ∫ x in face i, f (back_face i x) i)
```

5 Applications to complex analysis

The main goal of the project was to formalize a version of the divergence theorem that implies the Cauchy integral formula under standard assumptions. In this section, I will briefly explain how I deduce the Cauchy integral formula from Theorem 8. While many textbooks on complex analysis prove these formulas for functions $f\colon \mathbb{C} \to \mathbb{C}$, I prove them for functions that take values in a complex Banach space E. As before, I have to assume that E has a second countable topology because of the way the Bochner integral is defined in mathlib.

First, consider an open rectangle $R = \{ z \mid a \leqslant \operatorname{Re} z \leqslant b, c \leqslant \operatorname{Im} z \leqslant d \}$ on the complex plane $\mathbb{C}$. If $f\colon \mathbb{C} \to E$ is continuous on R and is complex differentiable at all but countably many points of the interior of R, then one can apply the divergence theorem to the function $F\colon \mathbb{C} \to E^2$ given by $F(z) = (-if(z), f(z))$. Due to Cauchy-Riemann equations, the left-hand side of (1) equals zero. It is easy to see that the right-hand side is equal to the integral $\int_{\partial R} f(z)\,dz$. Thus we have the Cauchy-Goursat theorem for a rectangular domain.

```
lemma cauchy_theorem_rect (f : ℂ → E) (z w : ℂ)
  (s : set ℂ) (hs : countable s)
  (Hc : continuous_on f (ICC z.re w.re ×ℂ Icc z.im w.im))
  (Hd : ∀ x ∈ (IOO z.re w.re ×ℂ IOO z.im w.im) \ s,
    differentiable_at ℂ f x) :
  (∫ x in z.re..w.re, f (x + z.im * i)) -
    (∫ x in z.re..w.re, f (x + w.im * i)) +
    (i · ∫ y in z.im..w.im, f (re w + y * i)) -
    i · ∫ y in z.im..w.im, f (re z + y * i) = 0 :=
```

To formulate the Cauchy-Goursat theorem for an annulus and a circle, I define the circle integral $\oint_{|z-c|=R} f(z)\,dz$ as $\int_0^{2\pi}(c+Re^{\theta i})'f(c+Re^{\theta i})\,d\theta$.

```
def circle_map\link{circle_map} (c : ℂ) (R : ℝ) : ℝ → ℂ := λ θ, c + R *
    exp (θ * i)

def circle_integral (f : ℂ → E) (c : ℂ) (R : ℝ) : E :=
∫ θ in 0..2 * π, deriv (circle_map c R) θ · f (circle_map c R θ)
```

Applying the Cauchy-Goursat theorem for a rectangle to the function $F(z) = e^z f(c + e^z)$ on the rectangle $\ln r \leqslant \operatorname{Re} z \leqslant \ln R$, $0 \leqslant \operatorname{Im} z \leqslant 2\pi$, I prove the Cauchy-Goursat theorem for a function differentiable on an annulus.

```
lemma cauchy_thm_annulus {c : ℂ} {r R : ℝ} (h0 : 0 < r)
  (hle : r ⩽ R) {f : ℂ → E} {s : set ℂ} (hs : countable s)
  (hc : continuous_on f (closed_ball c R \ ball c r))
  (hd : ∀ z ∈ ball c R \ closed_ball c r \ s, differentiable_at ℂ f z) :
  ∮ z in C(c, R), f z = ∮ z in C(c, r), f z
```

Next, I apply this theorem to the function $\frac{f(z)}{z-c}$ (formally, $(z-c)^{-1} \cdot f(z)$ because f is a vector-valued function) and take the limit as r tends to 0 from the right. Thus, I prove the Cauchy integral formula for the value of a complex differentiable function at the center of a disk. This theorem will be later generalized to any point of the disk, but I will use this case to prove the general version.

```
lemma cauchy_integral_disk_center {R : ℝ} (h0 : 0 < R)
  {f : ℂ → E} {c : ℂ} {s : set ℂ} (hs : countable s)
  (hc : continuous_on f (closed_ball c R))
  (hd : ∀ z ∈ ball c R \ s, differentiable_at ℂ f z) :
  ∮ z in C(c, R), (z - c)⁻¹ · f z = (2 * π * i : ℂ) · f c :=
```

Applying this lemma to the function $(z - c) \cdot f(z)$, we prove the Cauchy-Goursat theorem for a disk.

```
lemma cauchy_thm_disk {R : ℝ} (h0 : 0 ⩽ R) {f : ℂ → E}
  {c : ℂ} {s : set ℂ} (hs : countable s)
  (hc : continuous_on f (closed_ball c R))
  (hd : ∀ z ∈ ball c R \ s, differentiable_at ℂ f z) :
  ∮ z in C(c, R), f z = 0 :=
```

Next, I show the Cauchy integral formula

$$f(w) = \frac{1}{2\pi i} \oint\limits_{|z-c|=R} \frac{f(z)}{z-w}\,dz$$

for any point of the open disc, $|w-c| < R$. To obtain this result, I apply the Cauchy-Goursat theorem to the function

$$g(z) = \begin{cases} f'(w), & \text{if } z = w; \\ \frac{f(z)-f(w)}{z-w}, & \text{otherwise.} \end{cases}$$

If f is differentiable at w, then g satisfies the assumptions of the previous theorem, hence the integrals of $\frac{f(z)}{z-w}$ and $\frac{f(w)}{z-w}$ over the circle $|z-c| = R$ are equal. It is easy to see that the latter integral equals $2\pi i f(w)$.

If w belongs to the countable set where f is not guaranteed to be differentiable, then the same formula follows from the previous case by continuity.

```
lemma cauchy_integral_disk {R : ℝ} {c w : ℂ} {f : ℂ → E} {s : set ℂ}
  (hs : countable s) (hw : w ∈ ball c R)
  (hc : continuous_on f (closed_ball c R))
  (hd : ∀ x ∈ ball c R \ s, differentiable_at ℂ f x) :
  (2 * π * i : ℂ)⁻¹ • ∮ z in C(c, R), (z - w)⁻¹ • f z = f w :=
```

The Cauchy integral formula immediately implies that a function $f\colon \mathbb{C} \to E$ that is complex differentiable on an open disk and is continuous on the corresponding closed disk must be analytic on the interior of this disk. The coefficients of the Taylor series are given by Cauchy integrals.

```
def cauchy_power_series (f : ℂ → E) (c : ℂ) (R : ℝ) :
  formal_multilinear_series ℂ ℂ E :=
λ n, continuous_multilinear_map.mk_pi_field ℂ _
  ((2 * π * i : ℂ)⁻¹ • ∮ z in C(c, R), (z - c) ^ (-n - 1 : ℤ) • f z)
```

```
lemma analytic_of_differentiable {R : ℝ≥0} {c : ℂ} {f : ℂ → E}
  {s : set ℂ} (hs : countable s) (hc : continuous_on f (closed_ball c R))
  (hd : ∀ z ∈ ball c R \ s, differentiable_at ℂ f z) (hR : 0 < R) :
  has_fpower_series_on_ball f (cauchy_power_series f c R) c R :=
```

I started to use these theorems to build a complex analysis library in Lean. For example, I proved the Riemann removable singularity theorem, several versions of the maximum modulus principle, Liouville's theorem, and the Schwarz lemma.

```
lemma removable_singularity
  (hd : ∀ᶠ z in 𝓝[≠] c, differentiable_at ℂ f z)
  (ho : is_o (λ z, f z - f c) (λ z, (z - c)⁻¹) (𝓝[≠] c)) :
  ∃ y : E, tendsto f (𝓝[≠] c) (𝓝 y) :=
```

Here `𝓝[≠] c` is the filter of punctured neighborhoods of c.

```
theorem max_modulus (hU : bounded U) (hd : diff_cont_on_cl ℂ f U)
  (hC : ∀ z ∈ frontier U, ‖f z‖ ≤ C) (hz : z ∈ closure U) :
  ‖f z‖ ≤ C :=
```

Here `diff_cont_on_cl` is a predicate saying that a function is differentiable on a set and is continuous on its closure.

```
theorem liouville (hf : differentiable ℂ f) (hb : bounded (range f)) :
  ∃ c, ∀ x, f x = c :=
```

```
lemma schwarz_lemma (hd : differentiable_on ℂ f (ball 0 R))
  (h_maps : maps_to f (ball 0 R) (ball 0 R)) (h₀ : f 0 = 0)
  (hz : abs z < R) :
  abs (f z) ≤ abs z :=
```

6 Future plans

6.1 Other integrals

One possible direction of improvement would be to formalize the divergence theorem for (some of the) other integrals listed in the Bongiorno's survey [6]. Some of these theorems allow the function to be continuous, not differentiable, on a countable set of *coordinate hyperplanes*. These theorems can be transfered to more general versions of the Cauchy-Goursat theorems.

The main obstacle for this project is that I define "box" integrals in a non-extensible way, see Sec. 4.5. So, to add more integrals (or to replace the GP-integral with a better one), one has to seriously refactor the definition.

6.2 Complex analysis in higher dimension

Most of the proofs discussed in Sec. 5 can be easily generalized to the case of a function $f\colon \mathbb{C}^n \to E$. I mentor a student who tries to generalize the Cauchy integral formula as a part of a course.

6.3 The Cauchy-Goursat theorem for any domain

My goal was to show that it is possible to deduce the Cauchy-Goursat theorem from the divergence theorem, so I formalized the Cauchy-Goursat theorem only for a few special cases (a rectangle, an annulus, and a disk). One can deduce the general Cauchy-Goursat theorem from the version for a rectangle but it requires quite a few topological lemmas that are not in `mathlib` yet.

6.4 One-dimensional complex analysis

Formalization of the Cauchy-Goursat theorem makes it possible to formalize many theorems from the one-dimensional complex analysis. I already formalized a few theorems, but, clearly, this is just the beginning.

My main goal for the next year or two is to formalize Ilyashenko's proof of the fact that a polynomial vector field on $\mathbb{R}^2$ has only finitely many limit cycles (i.e., isolated periodic solutions), at least in the case of hyperbolic singular points. The proof of this theorem heavily relies on the complex analysis.

7 Conclusion

This project demonstrates that formalization of a sufficiently general version of the divergence theorem can be used to prove the Cauchy-Goursat theorem and bootstrap complex analysis. I hope that other systems will adopt this approach to formalization of the divergence theorem, or a similar one based on a better generalization of the Henstock-Kurzweil integral, see Sec. 6.1.

References

1 Mohammad Abdulaziz and Lawrence Paulson. An Isabelle/HOL formalisation of green's theorem. *J Autom Reasoning*, 63:763–786, 2019. doi:10.1007/s10817-018-9495-z.

2 Felipe Acker. The missing link. *The Mathematical Intelligencer*, 18(3):4–9, June 1996. doi:10.1007/BF03024304.

3 Jeremy Avigad, Johannes Hölzl, and Luke Serafin. A formally verified proof of the central limit theorem. *Journal of Automated Reasoning*, 59(4):389–423, 2017. doi:10.1007/s10817-017-9404-x.

4 Salomon Bochner. Integration von Funktionen, deren Werte die Elemente eines Vektorraumes sind. *Fundamenta Mathematicae*, 20, 1933.

5 Sylvie Boldo, François Clément, and Louise Leclerc. A Coq formalization of the Bochner integral, 2022. arXiv:2201.03242.

6 Benedetto Bongiorno. The Henstock-Kurzweil integral. In E. PAP, editor, *Handbook of Measure Theory*, chapter 13, pages 587–615. North-Holland, Amsterdam, 2002. `doi:10.1016/B978-044450263-6/50014-2`.

7 Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris van Doorn, and Jakob von Raumer. The Lean theorem prover (system description). In Amy P. Felty and Aart Middeldorp, editors, *Automated Deduction - CADE-25*, pages 378–388, Cham, 2015. Springer International Publishing.

8 Russel A. Gordon. *The integrals of Lebesgue, Denjoy, Perron, and Henstock*, volume 4 of *Graduate Studies in Mathematics*. American Mathematical Society, Providence, R.I, 1955.

9 John Harrison. Formalizing basic complex analysis. In R. Matuszewski and A. Zalewska, editors, *From Insight to Proof: Festschrift in Honour of Andrzej Trybulec*, volume 10(23) of *Studies in Logic, Grammar and Rhetoric*, pages 151–165. University of Białystok, 2007. URL: `http://mizar.org/trybulec65/`.

10 John Harrison. Formalizing an analytic proof of the prime number theorem. *Journal of Automated Reasoning*, 43(3):243–261, 2009. `doi:10.1007/s10817-009-9145-6`.

11 The mathlib community. The Lean mathematical library. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs*, CPP 2020, pages 367–381, New York, NY, USA, 2020. Association for Computing Machinery. `doi:10.1145/3372885.3373824`.

12 Jean Mawhin. Generalized multiple Perron integrals and the Green-Goursat theorem for differentiable vector fields. *Czechoslovak Math. J.*, 31(106)(4):614–632, 1981.

13 Washek F. Pfeffer. *The Riemann approach to integration*, volume 109 of *Cambridge Tracts in Mathematics*. Cambridge University Press, Cambridge, 1993. Local geometric theory.

14 Floris van Doorn. Formalized Haar Measure. In Liron Cohen and Cezary Kaliszyk, editors, *12th International Conference on Interactive Theorem Proving (ITP 2021)*, volume 193 of *Leibniz International Proceedings in Informatics (LIPIcs)*, pages 18:1–18:17, Dagstuhl, Germany, 2021. Schloss Dagstuhl – Leibniz-Zentrum für Informatik. `doi:10.4230/LIPIcs.ITP.2021.18`.